
Статические и динамические объекты. Ссылочный тип в языке 
Паскаль. 

Во многих языках программирования  объекты программы (переменные, 
константы, функции и др.)  могут быть статическими или динамическими. Здесь 
мы ограничимся рассмотрением одного вида программных объектов – 
переменных. Примером статического объекта  в языке Паскаль является 
переменная, описанная в блоке  программы или подпрограммы (процедуры, 
функции). Так, описание  

var n : integer; 
порождает статическую переменную целого (или целочисленного) типа. 
Отметим три важных момента, связанных с описанием переменной: 

1) Переменная получает имя (в нашем примере это имя  n), с помощью 
которого она изображается в программе. Другими словами, данное имя 
позволяет обращаться к переменной в программе1.  

2) Тип переменной задаёт множество допустимых для неё значений. Для  
переменной n это целые числа из диапазона от – Maxint до + Maxint, 
где Maxint – константа, определяемая реализацией языка Паскаль. 

3) Тип переменной задаёт её размер, т.е. объём машинной памяти, 
необходимый для размещения в этой переменной значений данного типа. 
Размер значений того или иного типа определяется конкретной 
реализацией языка. Так, в системе Турбо Паскаль  все значения типа 
integer в машинном представлении имеют одинаковый размер – два 
байта (или 16 двоичных разрядов). 

Отвлекаясь от машинного представления, переменную можно считать 
ячейкой или ящиком, который предназначен для хранения значений 
определённого типа, причём в каждый момент в ящике хранится не более 
одного значения. 

 Для наглядного представления характера значений и действий над 
переменными договоримся о следующей схеме их графического изображения. 
Переменные целого типа  и других простых типов  будем изображать на 
рисунках в виде прямоугольников. Слева от фигуры, изображающей 
переменную,  указывается её имя. Для целочисленной переменной n рисунок 
будет таким:  
 
  

В начальный момент выполнения программы значение переменной не 
определено. Этот факт изображается  пустым прямоугольником. Значение, 
присвоенное переменной в процессе  выполнения программы, будем вписывать 
внутрь соответствующего прямоугольника. Например, после выполнения 
оператора  n:=32,  рисунок изменится так:   
    

                                                           
1 Переменная и ее имя – это две разные сущности. Вместо полной фразы  «переменная, 
имеющая имя n , порождённая данным  описанием var n : integer»  будем использовать 
более короткую – «переменная n» (там, где это не вызовет недоразумений). 

n  

n 32 

Массивы и записи будем изображать фигурами, составленными из 
фигур, изображающих их компоненты и  расположенных соответственно одна 
под другой по вертикали или одна рядом с другой по горизонтали. Ниже 
показаны порожденные описанием  
var a : array[1..5] of integer;  
    b : record x: array[1..3] of integer; y,z: real end; 
массив из пяти чисел и запись с тремя полями. В записи первое поле – массив из 
трех целых чисел, второе и третье поля – вещественные числа. Имена a и b 
обозначают объекты целиком. Для указания компонент сложных объектов 
используются более сложные обозначения, например a[3], b.z или 
b.x[2]. 
   
 
 
 
 
 

Статические программные объекты  порождаются  автоматически 
перед выполнением программы  или подпрограммы, в которой они описаны, и 
существуют, пока выполнение этой программы или подпрограммы не 
завершится. Размер статических объектов не изменяется на протяжении всего 
времени их существования.  

К динамическим относятся объекты, которые с помощью специальных 
действий создаются (или изменяют свой размер) уже в процессе выполнения  
программы (подпрограммы). 

В языке Паскаль для создания динамических объектов используется 
стандартная процедура new, имеющая один параметр. Объект, созданный с 
помощью new в процессе выполнения программы или подпрограммы, 
существует вплоть до завершения основной программы, или до тех пор, пока он 
не будет уничтожен явно с помощью другой стандартной процедуры – 
dispose. 

Для того, чтобы в  программе на Паскале обращаться к статическому 
объекту, мы используем имя, данное ему в разделе описаний. Динамические  
объекты заранее не описываются – как же к ним обращаться? Для этого вместе с 
порождаемым объектом создается специальное значение – ссылка на этот 
объект. Чтобы хранить ссылочные значения, используются переменные 
специального ссылочного типа. При порождении объекта ссылка на него 
помещается в переменную, указанную в качестве фактического параметра 
оператора процедуры new. Каждому  динамическому объекту соответствует 
своя ссылка, различные объекты имеют различные  ссылки.  

На рисунках будем изображать ссылку точкой и связывать ее стрелкой с 
соответствующим объектом: 
 
 

В машинных терминах ссылка – это адрес объекта в оперативной памяти 
компьютера.  Язык Паскаль не позволяет явно записывать  адреса в программе, 
к тому же на этапе составления программы не известно, по какому адресу будет 

a  
 
 
 
 

a[3]

b  
 
 

  

b.z b.x[2]

расположен тот или иной динамический объект. Поскольку ссылки не имеют 
явного обозначения в программе, работа с динамическими объектами 
происходит посредством  статических переменных ссылочного  типа.  

Синтаксис задания ссылочного типа определяется следующим правилом 
БНФ: 
<задание ссылочного типа>::= ↑<имя типа> , 
где стрелка означает, что задаётся ссылочный тип, а <имя типа> – это имя  
любого стандартного или ранее описанного типа. 

Значениями переменных ссылочного типа могут быть ссылки на 
динамические объекты, причём только того типа, имя которого указано в 
задании после стрелки.  Переменные ссылочного типа описываются обычным 
способом. Например, в силу  описаний  
type ref_integer = ↑integer; 
var p : ref_integer; 
значением переменной p может быть ссылка на динамический объект целого 
типа.1 В начальный момент выполнения программы переменная p не имеет 
никакого значения (значение не определено) : 
 
 
Если далее с помощью оператора new(p)  порождается динамический объект, 
ссылка на него автоматически присваивается  переменной p.  Схематично 
результат изображается следующим образом: 
 
 
 

Можно сказать, что переменная p теперь "указывает" на объект целого типа. 
Поэтому переменные ссылочного типа часто называют указателями. Заметим, 
что параметр процедуры new однозначно определяет, какого типа объект 
порождается. В данном случае из описания типа переменной p следует, что 
порождается объект типа integer. Отметим также, что порождаемые объекты 
не имеют никакого начального значения.  

Среди значений любого ссылочного типа есть специальное значение, 
которое называется пустая ссылка. Оно не связано ни с каким объектом, т.е. ни 
на что фактически не ссылается. В программе такое значение для всех 
ссылочных типов явно изображается служебным словом nil. 

Пусть  q – ссылочная переменная того же типа, что и переменная p. 
Рисунок 
 
 
говорит о том, что значением q является пустая ссылка, то есть в данный 
момент  q не указывает ни на какой объект (пустой прямоугольник означал бы, 
что значение q  не определено). 

                                                           
1 Переменную ссылочного типа можно также описать, непосредственно используя задание типа 
в разделе описания переменных, например  var p :↑integer. 

     динамический объект  ( переменная типа integer ) 

p  

p  
 

q nil 

Осталось выяснить, как в программе работать с порождённым 
динамическим объектом, например, как присвоить ему значение. Если к 
обозначению указателя приписать справа стрелку ↑, то мы получим 
обозначение объекта, ссылка на который хранится в данном указателе. 
Выполнение оператора  p↑:= 35 приведёт к тому, что целая переменная, на 
которую указывает  p,  получит значение 35: 
   

значение p↑ до присваивания не 
определено 

 значение p↑ после присваивания  
равно 35 

 
Пусть n – имя переменной типа integer. Тогда при выполнении оператора 
n:=p↑ значение динамической переменной будет присвоено статической 
переменной n: 
 
   

значение n до присваивания не 
определено 

 значение n после присваивания  
равно 35 

 
Итак, приписывание стрелки справа к указателю – это способ изобразить 

динамическую переменную  в программе. Такое обозначение (со стрелкой 
справа) называется переменная с указателем. С помощью металингвистических 
формул (БНФ) подытожим все возможные обозначения переменных в языке 
Паскаль: 
<переменная>::= <полная переменная> | <переменная с 

индексом> | <компонента записи> | <буферная переменная> | 
<переменная с указателем> 

<полная переменная>  ::= <имя переменной> 

<переменная с индексом> ::= <переменная-массив> [ 
<индексное выражение> {,<индексное выражение>}] 
<компонента записи> ::= <переменная-запись>.<имя поля>| 

<имя поля> 

<буферная переменная>   ::= <файловая переменная>↑ 

<переменная с указателем> ::= <ссылочная переменная>↑ 

<переменная-массив>    ::=  <переменная>типа 'массив' 

<переменная-запись>    ::=  <переменная>типа 'запись' 
<файловая переменная>  ::=  <переменная>типа 'файл'    
<ссылочная переменная> ::=  <переменная>типа 'ссылка'   

<имя переменной>::=<идентификатор> 

p  
35 

p  
35 

n  

p  
 {p↑:= 35} 

p  
35 

{ n:=p↑ }

n 35 

 Заметим, что возможны случаи, когда синтаксически верное 
обозначение переменной не имеет смысла. Тогда  программа, в которой  
встретилось такое обозначение, не может быть исполнена (или её исполнение 
приведёт к ошибке). Например, обозначение a[i]ошибочно, если имя a в 
программе не означает массив. Если это имя описано с помощью var a : 
array[1..5] of integer,  а значение индекса  i во время исполнения 
программы равно 6, то обращение к a[i] приведёт к ошибке, так как не 
существует компоненты массива с таким индексом.  

Следует проявлять осторожность и при работе с указателями. Если, 
например, значение указателя q не определено или равно nil, то обращение  
к q↑ приведёт к ошибке. 

 Теперь рассмотрим операции над ссылочными значениями одного и того 
же типа. В Паскале ссылки можно присваивать и сравнивать на равенство и 
неравенство. Присваивание осуществляется, как обычно, с помощью оператора 
присваивания: 
 <ссылочная переменная> := <ссылочное выражение>  , 
где  <ссылочная переменная> – обозначение переменной ссылочного 
типа,  <ссылочное выражение> – это либо пустая ссылка nil, либо 
ссылочная переменная,  либо вызов функции ссылочного типа. 

Пусть переменные p и q имеют значения, изображенные на схеме:  
 
 
 
 
 
Тогда логические выражения  p=q и q<>nil дают значение "ложь". После 
присваивания q:=p (результат изображен на рисунке)    
   

значение q до присваивания 
равно nil  

 значение q  после присваивания 
совпадает со значением p, т.е. q 
ссылается на тот же объект, что и p 

логические выражения  p=q и q<>nil дают значение "истина". 
 
  Если объект, созданный с помощью оператора  new, больше не нужен 
программе, его целесообразно уничтожить, чтобы он зря не занимал место в 
оперативной памяти (на его месте могут быть созданы другие объекты).  

Уничтожение динамического объекта осуществляется оператором 
процедуры dispose(<ссылочное выражение>). Значением выражения, 
указанного в качестве фактического параметра должна быть ссылка на уже 
существующий динамический объект (иначе ошибка). Этот объект перестает 
существовать, a ссылка на него удаляется из множества  значений  ссылочного 

p  
35 

q nil 

p  
35 

q  
{ q:=p } p  

35 
q nil 

типа, в результате чего все указатели, содержащие ссылку на уничтоженный 
объект, становятся неопределёнными. 

Для  переменных p и  q, изображённых на последней схеме, выполнение 
оператора dispose(q) приведёт к исчезновению динамического объекта со 
значением  35,  а значения переменных  p и  q  станут неопределёнными. 
   

значения p и q определены и 
равны между собой  

 значения p и q ( оба! )  не 
определены  

Линейные списки и их реализация посредством статических и 
динамических структур данных 

Понятие списка хорошо известно из жизненных примеров: список 
студентов учебной группы, список призёров олимпиады, список (перечень) 
документов для представления в приёмную комиссию, список почтовой 
рассылки, список литературы для самостоятельного чтения и т.п. 

 В математике список (или кортеж) – это конечная последовательность 
(допускающая повторения) элементов какого-нибудь множества Ψ. Список 
обозначается <x1,x2,…,xn>, где n (n≥0) – количество элементов, или длина 
списка, для i=1,…,n xi есть i-й элемент списка (xi ∈ Ψ). Об элементе  xi 
говорят также, что он занимает  i-ю позицию в списке. При n = 0 получается 
пустой список, который не содержит элементов и обозначается < >. 

Элементы множества Ψ могут иметь весьма сложную структуру, отражая 
реальные объекты и процессы. Вследствие этого возможны списки, в которых 
некоторые элементы сами являются списками или содержат списки внутри себя. 
Такие списки называют иерархическими. В отличие от них списки, не 
допускающие таковых внутри себя, называются  линейными. 

Далее в качестве  множества Ψ будем рассматривать множество значений 
некоторого  типа языка Паскаль. Этот тип в общем случае будем обозначать 
именем elemtype (от англ. element type – тип элемента). 

Важной структурной особенностью списка является то, что его элементы 
линейно упорядочены в соответствии с их позицией в списке. Для  i=1,…,n-1 
элемент xi предшествует  элементу xi+1; для  i=2,…,n элемент xi  следует 
за элементом xi-1.  

Следующие операции можно применять к  линейным спискам: 

1) Получить значение i-го элемента списка или изменить i-й элемент; 
2) Напечатать элементы  списка в порядке их расположения; 
3) Найти в списке элемент с заданным значением; 
4) Определить длину списка; 

p  

q  

p  
35 

q  
{dispose(q)}

5) Вставить новый элемент непосредственно за i-м элементом или перед ним, 
вставить элемент в пустой список; 
6) Удалить i-й элемент; 
7) Соединить два линейных списка в один список; 
8) Разбить список на два списка; 
9) Создать копию списка; 
10) Сделать список пустым. 

Возможны и более сложные операции над линейными списками. 

Примеры. 

1. Вставить в список целых чисел <1,3,4,2,1,5> число 7 после третьего 
элемента.  

Результат:  <1,3,4,7,2,1,5>.  

В полученном  списке первые три элемента не изменились, четвертую позицию 
заняло вставляемое число 7, остальные элементы сдвинулись на одну позицию 
вправо. Длина списка увеличилась на единицу. 

2. Вставить строку "фиолетовый" в конец списка строк              

<"красный", "оранжевый", "жёлтый", "зелёный", "голубой", 
"синий">.     

Результат:      <"красный", "оранжевый", "жёлтый", "зелёный",   
"голубой", "синий", "фиолетовый"> 

3. Удалить из списка символов химических элементов <Ag,Au> первый 
элемент.  

Результат: <Au>. 

Длина списка уменьшилась на единицу. Если к результату применить ту же 
операцию – удалить первый элемент – получим пустой список < >.  

4. Вставить в пустой список вещественных чисел  < >  число 3.1415927.  

 Результат:    <3.1415927>. 

 Если в начало полученного списка вставить число 2.7182818, получим 
список <2.7182818,3.1415927>. 

5. Из списка названий месяцев удалить повторные названия: <December, 
January, February, March, April, May, December, May, May>.  

Результат: <December, January, February, March, April, May>. 

Реализация списков на Паскале 

Списки представляют собой удобную структуру данных для решения 
многих практических задач. Они используются, например, в программах 

информационного поиска, трансляторах, при моделировании различных 
процессов. Однако по отношению к языку Паскаль список является 
абстрактным понятием: в Паскале не предусмотрены ни встроенные  операции 
над списками, ни средства для их описания, подобные тем, что имеются, 
скажем, для массивов.1  Поэтому для обработки списков на Паскале потребуется 
их программная реализация. Для этого необходимо:  

1) сконструировать средствами Паскаля структуру данных, которая будет 
представлять в программе список (в этой структуре будут храниться 
элементы списка); 

2) описать в виде процедур и функций требуемые  операции над списками. 

 Мы рассмотрим два наиболее типичных способа реализации списков: 
посредством массивов и с помощью цепочки звеньев. 

Реализация списков посредством массивов  

 В этой реализации элементы списка располагаются по порядку в 
компонентах массива, начиная с первой; i-й элемент списка хранится в  i-ой 
компоненте. Размер массива определяет максимально возможную длину списка. 
В программе размер массива зададим константой maxlen. Для работы со 
списком требуется ещё знать его текущую длину (или номер компоненты 
массива, в которой находится текущий последний элемент). Опишем тип list 
(список) как запись, имеющую два поля:  первое с именем elems (от англ. 
elements – элементы) – собственно массив; второе с именем last – позиция 
последнего элемента списка. Дадим также описание переменной L типа list: 

const  
  maxlen =   … {подходящее для конкретной задачи целое  
                                                  число}; 
type  
  elemtype = … {подходящий для задачи тип элементов}; 

  list = record 
           elems: array [1..maxlen] of elemtype; 
           last: integer 
         end; 
var L: list; 

Выражение L соответствует понятию «список». Конструкция  L.elems[i] 
обозначает в программе i -й элемент списка.  

Пусть в качестве elemtype задан перечислимый тип: (Monday, 
Tuesday, Wednesday, Thursday, Friday, Saturday, Sunday). 
Реализация списка названий дней недели < Monday, Wednesday, Friday, 
Saturday> на схеме будет выглядеть так (считаем, что maxlen = 7) : 
                                                           
1 В  языке Лисп, напротив, список является основной структурой данных с богатым набором  
операций, а массивы  отсутствуют. 

Создать такой список можно следующими действиями: сделать список L 
пустым, затем поочередно вставлять в него элементы в нужном порядке. 
Сделать список пустым весьма просто – достаточно обнулить поле last, 
поскольку список пуст, если его длина равна нулю. Оформим это действие в 
виде процедуры  make_null(L): 

 procedure make_null(var L: list); 
 { делает список L пустым } 
  begin 
    L.last:=0 { длина списка стала равна нулю } 
  end; {make_null} 

Вставка нового элемента x  перед i-м элементом списка осуществляется в 
два этапа: во-первых, все значения из компонент массива elems с индексами i, 
i+1, …, last  перемещаются соответственно в компоненты с индексами i+1, i+2, 
…, last+1;  во-вторых,  в i-ю компоненту помещается элемент x,и значение 
поля last увеличивается на единицу.  

Ниже показано, что происходит при вставке значения    Thursday  
перед третьим элементом списка < Monday, Wednesday, Friday, 
Saturday>. 

после 1-го этапа вставки:         после 2-го этапа вставки: 

  

 
Операцию  вставки оформим в виде процедуры insert(L,x,i).  

procedure insert(var L: list; x: elemtype; i:integer); 
 { вставляет в список L элемент x перед  i-м элементом } 
 var p: integer; 
  begin 

в 3-ей компонен-
те теперь новое 
значение 

L Monday
Wednesday

Thursday
Friday

Saturday

5 
 3-я компонента свобо-
дна для записи  нового 
значения 

L Monday 
Wednesday 
Friday 
Friday 
Saturday 

 
 

4 

бывшие третий и 
четвертый элементы 
теперь занимают 
четвертую и пятую 
компоненты 

L Monday
Wednesday
Friday
Saturday

4 

    for  p:=L.last  downto  i  do  
      {перемещение значений из компонент i,i+1,…,last на  
       одну компоненту к концу массива} 
           L.elems[p+1]:=L.elems[p]; 
    L.elems[i]:=x; 
    L.last:=L.last+1; {длина списка увеличивается на 1} 
  end; {insert}  

Итак, чтобы вставить значение Thursday перед третьим элементом списка L, 
в программе следует использовать оператор insert(L, Thursday,3). 

Процедуру insert можно использовать и для вставки элемента в пустой 
список, задав единицу в качестве третьего параметра.  Например, 
make_null(L); insert(L, Sunday,1): 

{make_null(L)}  

  

  

Теперь покажем, как создать наш исходный список < Monday, 
Wednesday, Friday, Saturday >. Такой список получится в результате 
выполнения следующей последовательности операторов: make_null(L); 
insert(L, Saturday,1); insert(L, Friday,1); insert(L, 
Wednesday,1); insert(L, Monday,1). 

Для удаления i-го элемента опишем процедуру delete(L,i). 

procedure delete(var L: list; i:integer); 
 { удаляет из списка L i-й элемент } 
 var p: integer; 
  begin 
    for  p:=i  to  L.last-1  do  
      {перемещение значений из компонент i+1, 
       i+2,…,last на одну компоненту к началу   
       массива} 
           L.elems[p]:=L.elems[p+1]; 
    L.last:=L.last-1;{длина списка уменьшается на 1} 
  end; {delete}  

При использовании процедур insert и delete следует проявлять 
осторожность: нельзя вставлять элемент, если длина списка максимальна, т.е. 

L  
 

 

0 

… {insert(L, 
Sunday,1)} 

L Sunday 
 

 

1 

… 

равна maxlen; нельзя удалять из пустого списка; необходимо правильно 
задавать номер i  элемента в списке: 1 ≤ i ≤ last. Проверку условий для 
корректной работы со списками реализуем в виде логических функций: 

function is_full(var L: list):boolean; 
 { возвращает  истину, если длина списка максимальна, 
   иначе – ложь} 
  begin 
    is_full:=L.last=maxlen 
  end; {is_full}  

function is_empty(var L: list):boolean; 
 { возвращает  истину, если список пуст, иначе – ложь} 
  begin 
    is_empty:=L.last=0 
  end; {is_empty}  

function is_valid(var L: list;i:integer):boolean; 
 { возвращает  истину, если в списке есть i-й элемент,  
   иначе - ложь } 
  begin 
    is_valid:=(1<=i) and (i<=L.last) 
  end; {is_valid}  

Опишем также логическую функцию is_present(L, x), проверяющую, 
есть ли элемент x в списке L. 

function is_present(var L: list; x: elemtype):boolean; 
 { возвращает  истину, если элемент со значением x  
   присутствует в списке, иначе – ложь} 
var found:boolean; 
        p:integer; 
  begin  
    found:= false; 
    p:=1; 
    while not found and (p <= L.last) do 
      begin  
       found:= x = L.elems[p]; 
       p:= p+1  
      end; 
    is_present:= found 
  end; {is_present}  

Рассмотрим работу этой функции для разных случаев. Если список L пуст 
(L.last=0), цикл в теле функции не выполнится ни разу, так как условие 
p<=L.last ложно, и функция возвращает значение «ложь». Если элемент со 
значением x присутствует в списке, то переменная found после нескольких 
повторений тела цикла получит значение «истина», и цикл завершится, так как 

условие not found  станет ложным. Функция возвращает значение «истина». 
Наконец, если элемент отсутствует в непустом списке, то значение переменной 
found в цикле не изменится. Цикл завершится, поскольку на каждой итерации 
значение p увеличивается, и условие p<=L.last станет ложным. Значением 
функции будет «ложь». 

Отметим, что рассмотренная реализация списков посредством массивов 
позволяет достаточно легко просматривать содержимое списка (есть 
возможность непосредственного доступа к элементу по его индексу) и вставлять 
новые элементы списка в его конец. Однако, при вставке элемента в середину 
или при его удалении требуется перемещение всех последующих элементов. 
Кроме того, длина списка ограничена  константой maxlen. При достаточно 
больших значениях maxlen нерационально используется оперативная память, 
так как размер переменной, представляющей список, пропорционален константе  
maxlen, а не фактической длине списка. Таким образом, пустой список 
занимает столько же места в оперативной памяти, что и список максимальной 
длины.  

Реализация списков посредством цепочек динамических объектов 

Другой подход к реализации списков на языке Паскаль заключается в 
следующем. Для хранения отдельного элемента списка создается  динамический 
объект – запись из двух полей, называемая звеном. В одном из полей 
(информационном)  располагается сам элемент, а другое поле содержит ссылку 
на  звено, содержащее следующий элемент списка, или пустую ссылку, если 
следующего элемента нет. С помощью ссылок звенья как бы сцеплены в 
цепочку. Зная ссылку на первое звено можно добраться и до остальных звеньев, 
то есть ссылка на первое звено задаёт весь список. Пустой список 
представляется пустой ссылкой. Приведём  соответствующие описания на языке 
Паскаль. 
type link=↑node; { ссылка на звено } 
elemtype = … {подходящий для задачи тип элементов}; 
node= record           {звено состоит из двух полей:} 
   elem: elemtype; {элемент списка} 
   next: link      {ссылка на следующее звено} 
      end; 
list=link; {список задаётся ссылкой на звено } 
var L: list;{список} 

Обратим внимание на то, что в описании типа link используется имя 
node, а в описании типа node используется имя link. По правилам Паскаля в 
задании ссылочного типа можно укзывать имя, которое ещё не описано. 
Поэтому тип link описывается раньше, чем node. 

  Пример. Список символов <'a','b','c'>, представленный цепочкой 
звеньев, изображается так (в переменной L – ссылка на первое звено): 

При этом в программе выражение L означает ссылку на первое звено в цепочке; 
L↑ означает само первое звено, L↑.elem – первый элемент списка, L↑.next 
– ссылка на второе звено. Далее,  
L↑.next↑ – само второе звено, 
L↑.next↑.elem – второй элемент списка, 
L↑.next↑.next – ссылка на третье звено, 
L↑.next↑.next↑ – само третье звено, 
L↑.next↑.next↑.elem – третий элемент списка, 
L↑.next↑.next↑.next – пустая ссылка (конец списка). 
Выражение L имеет и другой смысл. Оно обозначает список  в программе, 
поскольку, зная ссылку на первое звено, мы имеем доступ ко всем остальным 
звеньям, т.е. «знаем»  весь список. С этой точки зрения выражение L↑.next в 
нашем примере обозначает список1 <'b','c'>, а выражение 
L↑.next↑.next↑.next – пустой список. 

Подчеркнём, что соседние звенья цепочки располагаются в  оперативной 
памяти произвольно относительно друг друга, в отличие от соседних компонент 
массива, всегда занимающих смежные участки памяти. Такое расположение 
звеньев облегчает операции вставки и удаления, так как нет необходимости 
перемещать элементы, как это было в случае реализации списков массивами. 
Поясним это на примерах. 

Пусть из списка <'a','b','c'>, представленного в программе 
переменной L, требуется удалить второй элемент. Для этого достаточно 
исключить из цепочки второе звено – «носитель» второго элемента. Изменим 
ссылку в поле next первого звена: L↑.next:= L↑.next↑.next. Теперь 
после первого звена в цепочке идёт звено, содержащее элемент 'c'. 
Получился список <'a','c'>. Чтобы исключённое звено не занимало места в 
памяти, его следует уничтожить процедурой dispose, предварительно 
запомнив ссылку на него во вспомогательной переменной q. Итак, 
окончательное решение таково: 

 q:=L↑.next; L↑.next:= L↑.next↑.next; dispose(q). 
Покажем на  рисунке происходящие после каждого действия изменения. 

 
 
 
 
 
 
 
 
 
                                                           
1 По правилам Паскаля имена link и list обозначают один и тот же тип, и мы вправе 
рассматривать L↑.next как переменную типа list,  означающую список. 

L  
'a'  

'b' 
'c' nil 

{q:=L↑.next} 

L  
'a'  

'b' 
'c' nil 

q  

Пусть теперь требуется вставить 'd' после первого элемента списка 
<'a', 'c'>. Решение состоит из двух этапов. Во-первых, необходимо создать 
«носитель» – звено для хранения вставляемого элемента, и занести этот элемент 
в поле elem «носителя». Во-вторых, путём изменения указателей включить 
созданное звено в цепочку после первого звена. Первый этап реализуется 
фрагментом new(q);q↑.elem:= 'd', где q – вспомогательная переменная 
типа link. Фрагмент q↑.next:= L↑.next; L↑.next:=q  осуществляет 
второй этап вставки. Следующий рисунок иллюстрирует этапы вставки. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

L  
'a'  'c' nil 

q  

{ L↑.next:= L↑.next↑.next } 

L  
'a'  

'b' 
'c' nil 

q  

{ dispose(q)}

L  
'a' 

'c' nil

'd' 
q  

{ new(q);q↑.elem:=’d’ } 

Из примеров видно, что длина цепочки (количество звеньев в ней) может 
динамически изменяться, т.е. изменяться в процессе выполнения программы. 
Подобно цепочкам можно сконструировать и более сложные структуры, в 
которых объекты связаны между собой с помощью ссылок. Такого рода 
структуры данных называются динамическими. 

Приведём реализацию некоторых операций над списками. 

function is_empty(L: list):boolean; 
 { возвращает  истину, если список пуст, иначе – ложь} 
  begin 
    is_empty:= L = nil 
  end; {is_empty}  

 

function is_valid(L: list; i:integer):boolean; 
 { возвращает  истину,если в списке есть i-й элемент,  
   иначе - ложь } 
 var p: link; k:integer; 
  begin 
  p:=L; k:=1; 
  while (p< > nil) and (k<=i) do 
    begin p:=p↑.next; k:=k+1 end; 
    is_valid:=(k>i) and (i>0) 
  end; {is_valid}  

Рассмотрим работу функции is_valid для разных случаев.  Если список L 
пуст (L = nil), цикл в теле функции не выполнится ни разу, так как условие 
p < > nil ложно. Значение переменной k равно единице, поэтому выражение 
(k>i) and (i>0) ложно для любого i. Функция возвратит значение «ложь». 
Если i<1,  то цикл также ни разу не выполнится, так как при начальном 
значении k, равном единице, отношение k<=i ложно. Отношение i>0 также 
ложно, следовательно, значением выражения (k>i) and (i>0) будет 
«ложь», т.е. функция возвращает значение «ложь». Если i>=1 и в списке есть 
i-й элемент, то тело цикла выполнится  ровно i раз. Переменная k получит 

L  
'a' 

'c' nil

'd' 
q  

{ q↑.next:= L↑.next; L↑.next:=q } 

значение i+1,  выражение (k>i) and (i>0) истинно, и функция возвращает 
значение «истина». Наконец, если i>=1 и в списке нет i-го элемента, то 
выход из цикла произойдёт из-за ложности выражения p < > nil (по 
достижении конца списка). При этом выражение k<=i  останется истинным. 
Следовательно, k>i ложно и (k>i) and (i>0)ложно. Значением функции 
будет «ложь». 

 function ref(L: list; x: elemtype):link; 
 { возвращает ссылку на звено, содержащее элемент x,если   
   x входит в список L, иначе – nil } 
var p: link; found: boolean; 
  begin 
    found:=false; p:=L; 
    while not found and (p< > nil) do 
      if x = p↑.elem then found:=true  
                    else p:=p↑.next; 
      ref:=p  ; 
  end; {ref}  

procedure insert(var L: list; x,y: elemtype); 
 { вставляет в список L элемент y после элемента x, если  
   x входит в список } 
 var p,q: link; 
  begin 
    p:=ref(L,x); 
    if p < > nil then  
      begin q:= p↑.next; 
            new(p↑.next) 
            p↑.next↑.elem:= y; 
            p↑.next↑.next:= q 
      end  
  end; {insert}  

procedure delete(var L: list; x elemtype); 
 { удаляет из списка L элемент, следующий за x, если  
   x входит в список и x – не последний элемент в списке} 
 var p,q: link; 
  begin 
    p:=ref(L,x); 
    if p< >nil then  
      begin q:=p↑.next; 
            if q<>nil then  
               begin p↑.next:= q↑.next; dispose(q) 
               end 
      end  
  end; {delete}  

procedure destruct(L: list); 
 { разрушает список L,освобождая память, занимаемую 
   звеньями } 
 var q: link; 
  begin 
    while L < > nil do 
      begin q:=L; 
            L:=L↑.next; 
            dispose(q) 
      end 
  end; {destruct}  

Формальный параметр процедуры destruct описан как параметр-
значение. Следовательно, на месте фактического параметра при обращении к 
этой процедуре может быть любое ссылочное выражение. Например, вызов 
функции, которая возвращает ссылку на первое звено списка или nil.  

Пусть переменные L1 и L2  имеют одинаковые значения – ссылку на 
первое звено некоторого списка. Чтобы сделать  список L1 (и L2) пустым, 
понадобится такой фрагмент: destruct(L1); L1:=nil; L2:=nil. После 
выполнения оператора destruct(L1) значения обеих переменных L1 и L2 не 
определены, но после операторов присваивания их значением становится пустая 
ссылка, т.е. L1 (и L2) представляют теперь пустой список. 

Как уже отмечалось, операции вставки и удаления при реализации 
списков цепочками требуют меньше действий, чем при реализации  массивами, 
так как не требуется перемещать элементы, следующие за вставляемым (или 
удаляемым). Поэтому данная реализация больше подходит для задач, в которых 
изменения в списках происходят часто. Кроме того, в ней нет ограничения на  
длину списка (максимальная длина зависит только от ресурсов вычислительной 
системы); эффективнее используется память (занимаемый объём памяти 
пропорционален текущей длине списка).  

Недостатками данной реализации являются необходимость хранить в 
звеньях дополнительную информацию –  ссылки, а так же отсутствие прямого 
доступа к i-му элементу списка. Если нужно присвоить переменной x значение 
i-го элемента списка L, то в случае реализации массивами это делается одним 
оператором x:=L.elems[i], а при реализации цепочками понадобится 
фрагмент, содержащий цикл: p:=L; for k:=1 to i-1 do p:=p↑.next; 
x:= p↑.elem. По этой причине в задачах, где списки меняются сравнительно 
редко, а просмотр элементов осуществляется часто, выгоднее использовать 
реализацию с помощью массивов. 

Цепочку динамических объектов, реализующую список, называют 
однонаправленным списком. В литературе по программированию под термином 
«список» часто подразумевается именно однонаправленный список. Данное 
значение термина обычно имеется в виду и тогда, когда употребляют 
выражение «звено списка». В некоторых задачах важно  уметь продвигаться по 
списку не только от начала к концу, но и  в обратном направлении. Для этого 
используются двунаправленные списки. Звено двунаправленного списка 
содержит две ссылки – на следующее и на предыдущее звенья. Эта особенность 

должна учитываться при реализации операций (например, вставки, удаления) 
для двунаправленных списков. Ещё один вариант реализации списков в виде 
динамических цепочек – так называемые списки с заглавным звеном – 
рассматривается в следующем подразделе. 

Списки с заглавным звеном 

Если в начало обычного однонаправленного списка добавить 
дополнительное звено, не содержащее элемента  (т.е. поле elem в этом звене не 
определено), то получим список с заглавным звеном. Пример: 

 
 
 
Для таких списков некоторые алгоритмы обработки записываются 

проще, так как в списке всегда есть хотя бы одно звено (даже если он пуст). 
Пусть надо реализовать операцию вставки в конец списка L элемента х. 
Сравним решения для списка без заглавного звена и для списка с заглавным 
звеном: 

procedure insert1(var L: list; x: elemtype); 
 { вставляет в конец списка L  без заглавного звена   
   элемент x} 
 var p,q: link; 
  begin 
    new(q); q↑.elem:= x; q↑.next:= nil; 
    {q указывает на созданное для вставляемого элемента  
      звено } 
    if L = nil then L:= q  {случай пустого списка  
                        обрабатывается отдельно}  
             else 
      begin p:= L; 
        while p↑.next< > nil  do p:=  p↑.next; 
         {p указывает на последнее звено в списке}  
        p↑.next:= q 
         {присоединили в конец списка новое звено} 
      end  
  end; {insert1}  

 

procedure insert2(L: list; x: elemtype); 
 { вставляет в конец списка L  с заглавным звеном   
   элемент x} 
 var p,q : link; 
  begin 
    new(q); q↑.elem := x; q↑.next:= nil; p:= L; 
    while p↑.next< > nil do p:= p↑.next; 
     {p указывает на последнее звено в списке}  

L    
'a' 

'b' nil 

    p↑.next := q {присоединили в конец списка новое звено} 
  end; {insert2}  

Как видим, для списка с заглавным звеном нет нужды рассматривать случай 
L = nil отдельно. 

Рекурсивная обработка списков 

Вспомним математическое определение списка: это конечная 
последовательность элементов (возможно, пустая). В непустой 
последовательности можно выделить первый элемент – «голову» 
последовательности –  и оставшуюся часть – «хвост». Заметим, что «хвост» в 
свою очередь также является последовательностью. Таким образом, получаем 
рекурсивное определение последовательности: 
<последовательность>::= <голова> <хвост> | <пусто> 

<голова>::=<элемент>  

<хвост>::=<последовательность> 

Опираясь на это определение, можно задать рекурсивную процедуру 
(функцию)1 поэлементной обработки последовательности. Пусть, например, 
требуется напечатать список символов L (тип элементов – char). Процедуру 
печати можно сформулировать так: если список пуст – ничего не делать (так как 
нечего печатать), иначе напечатать «голову», а за ней напечатать «хвост». Для 
печати «хвоста» используем эту же процедуру (рекурсивное применение). Ниже 
приводятся рекурсивные процедуры print1 и print2 для печати списков 
соответственно без заглавного звена и с заглавным звеном. 

procedure print1 (L: list); 
{ печать списка без заглавного звена } 
  begin if L <> nil then 
    begin  write(L↑.elem,' '); {печать «головы»} 
           print1(L↑.next)     {печать «хвоста»} 
    end; 
  end; {print1} 
 
procedure print2 (L: list); 
{ печать списка с заглавным звеном } 
  begin if L↑.next <> nil then 
    begin  write(L↑.next↑.elem,' '); 
           print2(L↑.next)  
    end; 
  end; {print2} 

 

                                                           
1 Процедуры (функции), которые во время исполнения могут прямо или косвенно (через другие 
процедуры или функции) обратиться сами к себе, называются рекурсивными. 

Рассмотрим ещё одну задачу: «Не используя операторов цикла и 
перехода, описать функцию sum(L), вычисляющую сумму элементов списка 
целых чисел L. Считать, что сумма пустого списка равна нулю, и что список 
представлен цепочкой звеньев без заглавного звена».  

Поскольку циклы запрещены условием задачи, для прохода по списку 
воспользуемся рекурсией (как в реализации процедуры print1). Дадим 
рекурсивное определение суммы списка. Сумма пустого списка равна нулю, 
сумма непустого списка есть результат сложения «головы» с суммой «хвоста»: 

 
0, если L пуст, 

sum(L)= 
  
L↑.elem+sum(L↑.next),иначе. 

Теперь запишем определение функции на Паскале. 

function sum(L:list):integer; 
begin 
  if L = nil then sum:= 0 
            else sum:= L↑.elem +   
                    sum(L↑.next) 
end;{sum} 

Параметр L передаётся  по значению. Поэтому при каждом вызове этой 
функции создаётся локальная (по отношению к данному вызову) переменная L, 
и ей присваивается значение фактического параметра. При выходе из функции 
локальная переменная исчезает, и значение функции возвращается в точку 
обращения к ней. При рекурсивном обращении могут одновременно 
существовать несколько (различных !) локальных переменных с именем L (по 
одной для каждого вызова). Рисунок 1 иллюстрирует выполнение оператора 
write(sum(Z))основной программы, где Z представляет список  <5,-3>. 

Очереди  и  стеки 

В повседневной жизни часто приходится иметь дело с хранилищами 
различных объектов. В библиотеке хранятся книги; в автоматическом оружии 
есть магазин, в котором хранятся патроны; в холле парикмахерской «хранятся» 
клиенты, ожидающие своей очереди на обслуживание. Хранилище может быть 
пустым, если в данный момент никаких объектов в нём  нет. Независимо от 
того, как устроено хранилище, и какие объекты в нём хранятся, можно выделить 
два основных действия для работы с ним: 1) добавить объект (положить на 
хранение); 2) взять объект (изъять из хранилища). Если потребовать, чтобы 
операции добавления и изъятия объектов удовлетворяли некоторым условиям, 
можно выделить два особых типа хранилищ: очередь и стек. 

В хранилищах типа «очередь» (или FIFO1) для каждого добавляемого 

объекта справедливо следующее: он не может быть изъят раньше, чем любой 

                                                           
1FIFO означает “First In – First Out”, т.е. «первым вошёл – первым вышел». 

возвращает 2 

возвращает -3 

возвращает 0

    … 
write(sum(Z)); 

 
function sum(L:list):integer; 
begin 
  if L=nil then sum:=0 
           else sum:=L↑.elem +   
                    sum(L↑.next) 
end;{sum} 

L  

 
function sum(L:list):integer; 
begin 
  if L=nil then sum:=0 
           else sum:=L↑.elem +   
                    sum(L↑.next) 
end;{sum} 

L  

 
function sum(L:list):integer; 
begin 
  if L=nil then sum:=0 
           else sum:=L↑.elem +   
                    sum(L↑.next) 
end;{sum} 

L nil

Z  
5 

-3 nil

Рис. 1 

другой объект, уже находящийся в данный момент на хранении. Пример – 
транспортёрная лента для погрузки багажа. Чемодан, раньше других попавший 
на ленту, раньше других сойдёт с неё. 

В хранилищах типа «стек» (или LIFO2) каждый добавляемый объект не 
может быть изъят позже, чем любой другой объект, уже находящийся в данный 
момент на хранении. Примеры: стопка тарелок на полке буфета; магазин в 
автоматическом оружии. В этих примерах соблюдается дисциплина LIFO, так 
как взять можно только верхний предмет (тарелку или патрон), а добавить 
новый объект можно, только положив его на верхний. 

Потребность в использовании стека или очереди возникает при 
программировании многих реальных процессов и ситуаций. Рассмотрим один 
из способов организации очередей и стеков в программе. В качестве структуры 
данных для хранения объектов (элементов) будем использовать список. В 
случае очереди  элементы добавляются в конец списка, а удаляются из его 
начала. В случае стека элементы добавляются и удаляются с одного конца 
списка. Такой способ представления стеков и очередей очень распространён, 
так что на его основе можно дать более узкие определения данных понятий: 

Очередь – это список, в который элементы всегда добавляются в конец, а 
удаляются из начала. 

Стек – это список, в котором все вставки и удаления выполняются на 
одном конце, называемом вершиной стека.  

Каждая из возможных реализаций списка на Паскале может быть 
использована для представления очередей и стеков. Для стеков рассмотрим 
вариант представления с помощью массивов, а для очередей – с помощью 
динамических цепочек. 

Представление стеков с помощью массивов 

Опишем тип данных stack и две процедуры: push(S,x) – положить в 
стек S элемент x и pop(S,x) – взять с вершины стека S элемент, присвоив его 
параметру x. 

const  
  stacksize = … {максимальное число элементов, которые  
                  могут одновременно находиться в стеке};  
type  
  elemtype = … {подходящий для задачи тип элементов  
                   стека}; 
  stack = record 
           elems: array [1..stacksize] of elemtype; 
           top: integer {индекс вершины стека} 
          end; 

var S: stack; {стек} 

                                                           
2LIFO означает “Last In – First Out”, т.е. «последним вошёл – первым вышел». 

procedure push (var S: stack;x:elemtype); 
{ положить x в стек S} 
  begin S.top:=S.top+1; 
        S.elems[S.top]:=x 
  end; {push} 

procedure pop (var S: stack; var x:elemtype); 
{ взять из стека S верхний элемент и присвоить его x} 
  begin x:=S.elems[S.top]; 
        S.top:=S.top-1; 
  end; {pop} 
 
Перед началом работы со стеком нужно сделать его пустым. Для этого опишем 
процедуру init_stack(S): 

procedure init_stack (var S: stack); 
{ начальная установка стека } 
  begin  S.top:=0; 
  end; {init_stack} 

К пустому стеку нельзя применять операцию pop, а к полному стеку нельзя 
применять операцию push. Опишем функции full_stack(S)и 
empty_stack(S), проверяющие стек на переполнение и пустоту. 
 

function full_stack(var S: stack):boolean; 
 { возвращает  истину, если стек полон, иначе – ложь} 
  begin 
    full_stack:=L.top=stacksize 
  end; {full_stack}  

function empty_stack (var S: stack):boolean; 
 { возвращает  истину, если стек пуст, иначе – ложь} 
  begin 
    empty_stack:=L.top=0 
  end; {empty_stack}  

В некоторых алгоритмах, использующих стек, проверка на пустоту или 
переполнение перед обращением к стеку не требуется, поскольку для них 
доказано, что при корректных входных данных  подобных ситуаций не 
возникает.1 Для отладки программ, построенных по таким алгоритмам, было бы 
удобно, чтобы процедуры pop  и push сами проверяли корректность операции 
со стеком и в случае ошибки сообщали о ней. Пусть процедура error с одним 
параметром – строкой – печатает эту строку и приводит к завершению 
программы. Дадим соответствующие описания push и pop  . 
                                                           
1 В задаче 9 раздела «Примеры задач с решениями» описан алгоритм, использующий стек, в 
котором при корректных входных данных попытка взять элемент из пустого стека исключена. 
См. также замечание в решении задачи 10 в том же разделе. 

procedure push (var S: stack;x:elemtype); 
{ положить x в стек S с контролем переполнения} 
  begin  
    if S.top=stacksize then error('стек полон') 
      else begin 
             S.top:=S.top+1; 
             S.elems[S.top]:=x 
           end 
  end; {push} 

procedure pop (var S: stack; var x:elemtype); 
{ если стек непуст, взять из стека S верхний элемент и  
  присвоить его x, иначе сообщить об ошибке} 
  begin  
    if S.top=0 then error('стек пуст ') 
      else begin 
             x:=S.elems[S.top]; 
             S.top:=S.top-1; 
           end 
  end; {pop} 

После того, как программа отлажена, проверки становятся ненужными и их 
следует исключить в целях повышения быстродействия программы. 

Заметим, что в случаях, когда максимальный размер стека не может быть 
заранее определён, лучше воспользоваться реализацией на основе динамических 
цепочек. Тогда размер ограничивается только ресурсами вычислительной 
среды. 

Представление очереди с помощью списка с заглавным звеном 

 Поскольку операции вставки и удаления осуществляются на разных концах 
очереди, для работы с ней удобно иметь два указателя – на заглавное и на 
последнее звенья. Если очередь пуста, оба указателя ссылаются на заглавное 
звено. Приведём необходимые описания. 

type link =↑node; { ссылка на звено } 
elemtype = … {подходящий тип элементов очереди}; 
node= record            
   elem: elemtype; {элемент } 
   next: link      {ссылка на следующее звено} 
      end; 
queue = record  
          front: link {ссылка на заглавное звено}; 
           rear: link {ссылка на последнее звено} 
        end;  

var Q: queue;{очередь} 

procedure init_queue(var Q: queue); 
{ начальная установка очереди } 
  begin  new(Q.front);{создание заглавного звена} 
         Q.front↑.next:=nil; 
         Q.rear := Q.front; 
  end; {init_queue} 

function empty_queue(Q: queue):boolean; 
 { возвращает  истину, если очередь пуста, иначе – ложь} 
  begin 
    empty_queue:= Q.rear = Q.front 
  end; {empty_queue}  

procedure enqueue(var Q: queue; x:elemtype); 
{поставить в очередь Q элемент х} 
 begin  
  new(Q.rear↑.next); 
  Q.rear:= Q.rear↑.next; 
  Q.rear↑.elem:=x; 

  Z 

{ enqueue(Z,1) } 

{ init_queue(Z) }  Z

 nil

{ enqueue(Z,2) } 

1  

 Z

 2 nil

{ dequeue(Z,y) } 

Z   

2 nil
1 

Z   

1 nil
 

Рис. 2а 

1y

  Q.rear↑.next:= nil 
 end; {enqueue} 

procedure dequeue(var Q: queue; var x:elemtype); 
{взять из очереди Q элемент и присвоить х} 
var p:link;  
 begin  
  x:= Q.front↑.next↑.elem; 
  p:= Q.front; 
  Q.front:= Q.front↑.next; 
  dispose(p) 
 end; {dequeue} 

Пусть в основной программе описаны переменные: Z – очередь целых 
чисел, y – целое. На рисунке 2а  показан результат последовательного 
применения команд init_queue(Z), enqueue(Z,1), enqueue(Z,2), 
dequeue(Z,y). На рисунке 2б  подробно изображён процесс выполнения 
последней команды. Как видим, после исключения из очереди единица 
оказалась в поле elem заглавного звена, перестав быть частью очереди. 

При необходимости контролировать корректность использования 
очереди во время отладки программы, в процедуру dequeue можно добавить 
обращение к  процедуре error при попытке взять элемент из пустой очереди. 

Примеры  задач с решениями 

В задачах 1-8 используются списки без заглавного звена при следующем 
их описании: 
type link=↑node; { ссылка на звено } 

elemtype = … {подходящий для задачи тип элементов}; 
node= record           {звено состоит из двух полей:} 
   elem: elemtype; {элемент списка} 
   next: link      {ссылка на следующее звено} 
      end; 
list=link; {список задаётся ссылкой на звено } 

 

Задача 1. Описать процедуру create(L), которая создаёт список L из строки 
текстового файла input. 

Решение 

procedure create(var L:list); 
{создает список из символов  строки текстового файла  
                                input; elemtype=char} 

var p: link;  
 begin 
 write('=>'); {выводим приглашение для ввода строки} 
 if eoln then {пустая строка - список пуст} 
              L:=nil 
   else 
     begin {создаем первое звено} 
      new(p); read(p↑.elem); 
      L:=p; 

{ dequeue(Z,y) } 

1  

 Z

 2 nil

 y

procedure dequeue(var Q: queue;  
                var x:elemtype); 
var p:link;  
begin       

x:= Q.front↑.next ↑.elem; 
p:=Q.front; 
 

Q.front:= Q.front↑.next; 
 

dispose(p) 

end; {dequeue} 

x
Q
p  

1  

 Z

 2 nil

1y

x
Q
p  

x
Q
p  

1  

 Z

  2 nil

1y

Z   

2 nil
1  

Рис. 2б 

      while  not eoln do 
        begin {добавляем в список следующее звено} 
         new(p↑.next);  read(p↑.next↑.elem ); 
         p:=p↑.next 
         {p указывает на добавленное звено } 
        end; 
      { последнее звено должно иметь ссылку nil } 
      p↑.next:=nil  
     end ; 
     readln {пропускаем маркер конца строки} 
 end; 

 

Задача 2. Описать процедуру print(L), которая печатает все элементы списка 
L. Тип элементов – char. 

Решение 

procedure print(L:list); 
{печатает в файл output строку из элементов списка L} 
begin 
 while L<>nil do 
  begin write(L↑.elem); 
        L:=L↑.next 
  end; 
 writeln 
end; 

Рекурсивное решение было приведено в разделе «Рекурсивная обработка 
списков». 

Задача 3. Описать процедуру exchange(L), которая меняет первый и 
последний элементы непустого списка L. 

Решение 

procedure exchange(L:list); 
{ меняет местами элементы первого и последнего звеньев 
  непустого списка L} 
var e: elemtype; 
    p: link; 
begin  
  p:=L; {находим последнее звено} 
  while p↑.next<>nil do p:=p↑.next; 
  {теперь p указывет на последнее звено; 
   L указывает на первое} 
  {меняем местами первый и последний элементы} 
    e:=L↑.elem; L↑.elem:=p↑.elem; p↑.elem:=e 
end; 



 

- 31 - 

Задача 4. Описать функцию equal(L1,L2), которая проверяет на равенство 
списки  L1  и  L2. 

Решение 

Если хотя бы один из списков пуст, то в качестве результата функции 
можно взять значение выражения L1 = L2. Действительно, если оба списка 
пусты (т.е. L1 = nil и L2 = nil), то они равны и выражение L1 = L2 даёт 
значение «истина»; если один пуст, другой не пуст (т.е. списки не равны), то 
значением выражения будет «ложь».  В случае, когда оба списка не пусты, 
нужно организовать цикл, в котором два указателя L1 и L2 будут синхронно 
«пробегать» по звеньям обоих списков, начиная с первых звеньев. Выход из 
цикла происходит, если достигнуто последнее звено хотя бы в одном  из 
списков (т.е. L1↑.next = nil или L2↑.next = nil) или  найдены звенья с 
различающимися элементами(L1↑.elem < > L2↑.elem). Если выход 
произошёл  из-за неравенства элементов, то результат функции – «ложь». Если 
же элементы равны, то результат зависит от того, указывают ли оба указателя 
L1 и L2  на последние звенья. Это можно проверить с помощью выражения 
L1↑.next = L2↑.next, которое принимает значение «истина» если и только  
если L1↑.next = nil   и  L2↑.next = nil. 

function equal(L1,L2:list): boolean; 
{возвращает истину, если списки L1 и L2 равны, иначе – 
ложь} 
begin  
  if (L1 = nil) or (L2 = nil) 
    then equal:= L1=L2 {истина, если и только если оба 
                       списка пусты} 
    else {оба списка непусты; L1 и L2 синхронно пробегают  
          звенья списков} 
      begin 
       while (L1↑.next<>nil) and (L2↑.next<>nil)  
                         and (L1↑.elem=L2↑.elem) 
       {пока не достигли последнего звена в каком-либо  
        из списков  и текущие элементы списков равны,  
        переходим в каждом из списков к следующему звену} 
          do begin L1:=L1↑.next; L2:=L2↑.next end; 
       equal:=(L1↑.next=L2↑.next) and (L1↑.elem=L2↑.elem) 
       {истина, если и только если L1 и L2 указывают на   
        последние звенья и элементы в этих звеньях равны} 
      end 
end; 

Задача 5. Описать процедуру reverse(L), которая переворачивает список L, 
то есть первый элемент списка становится последним, второй – предпоследним 
и т. д., бывший последним становится первым.  

Решение 

Для решения данной задачи достаточно изменить ссылки так, чтобы 
звенья в цепочке расположились в обратном порядке. Для этого в поле next 
первого звена следует записать nil, тем самым сделав его последним; в поле 
next второго звена записать ссылку на бывшее первое звено и далее 
аналогично поступать с остальными звеньями, пока не дойдём до последнего. 
Чтобы бывшее последним звено стало первым, надо присвоить ссылку на него 
переменной L, представляющей список. 

procedure reverse(var L:list); 
{ переворачивает список L, т.е. изменяет ссылки в этом  
  списке так, чтобы его элементы оказались расположенными  
  в обратном порядке } 
var p,q:link; 
begin 
  if L<>nil {если L=nil, результатом будет пустой список} 
    then 
     begin p:=L; q:=L↑.next; 
       L↑.next:=nil; {первое звено становится последним} 
       while q<>nil {пока есть следующее звено} do 
         begin L:=q; {перешли к новому звену; L – указы- 
               вает на текущее звено, p – на предыдущее } 
               q:=q↑.next; {запомнили ссылку на хвост  
                            списка } 
               L↑.next:=p; {соединили текущее звено с  
                            предыдущим} 
               p:=L  {запомнили ссылку на текущее звено} 
         end 
     end 
end; 

Задача 6. Описать процедуру in_order(L,x), которая в список L 
упорядоченных по неубыванию целых чисел вставляет элемент x, сохраняя 
упорядоченность.  

Решение 

procedure in_order(var L:list; x:elemtype); 
{вставляет x в список L  упорядоченных по неубыванию целых 
 чисел, сохраняя упорядоченность(elemtype=integer)} 
var p,q:link; 
begin 
{создать звено для хранения элемента} 
  new(p);  
  if L = nil then {вставка в пустой список} 
    begin p↑.elem:= x;  
          p↑.next:= nil; 
          L:= p  
    end 

   else 
    begin q:=L; {поиск подходящего места для вставки}  
      while (q↑.next< > nil) and (q↑.elem <= x)do 
                                  q:=q↑.next; 
      if q↑.elem <= x then {вставка в конец списка} 
        begin 
          p↑.elem:= x;  
          p↑.next:= nil; 
          q↑.next:= p  
        end 
       else  {вставка x перед q↑.elem} 
        begin  
          p↑.next:= q↑.next; 
          q↑.next:= p; 
          p↑.elem:= q↑.elem; 
          q↑.elem:= x 
        end 
    end  
end; 
   

 

Задача 7. Описать рекурсивную процедуру count(L,e), которая 
подсчитывает число вхождений элемента e в список L.  

Решение 

Если список пуст, то число вхождений элемента e в список L равно 
нулю. Для непустого списка нужно подсчитать, сколько вхождений имеется в 
«хвосте» списка, и прибавить единицу или ноль в зависимости от того, 
совпадает ли «голова» с элементом e. 

function count(L:list;e:elemtype):integer; 
{подсчитывает число вхождений элемента e в список L} 
begin 
  if L = nil then count:=0  
     else count:=count(L↑.next,e)+ord(L↑.elem=e) 
end; 

Задача 8. Описать рекурсивную функцию copy(L), которая строит копию 
списка L, возвращая ссылку на первое звено построенного списка или nil, если  
L  пуст.  

Решение 

function copy(L:list):list; 
{возвращает ссылку на копию списка L} 
var head:link; 
begin 

if L = nil then copy:= nil 
         else 
           begin  {создаем  копию «головы» списка} 
               new(head); 
               head↑.elem:= L↑.elem; 
               {к созданной «голове» присоединяем  
                копию «хвоста»} 
               head↑.next:= copy(L↑.next)  
           end 
end; 

Задача 9. Используя стек (тип данных и операции над стеком считать уже 
описанными), описать процедуру pairs(f) для решения следующей задачи. В 
текстовом файле f находится последовательность символов, сбалансированная  
по круглым скобкам. Требуется для каждой пары соответствующих 
открывающей и закрывающей скобок вывести на экран номера их позиций в 
последовательности, упорядочив пары номеров в порядке возрастания номеров 
позиций закрывающих скобок. Например, для текста  A+(45-F(X)*(B-C)) 
надо вывести: 8 10; 12 16; 3 17 . 

Решение 

Опишем в процедуре переменную sym для хранения очередного 
считанного из файла f символа; переменную pos для хранения номера позиции 
текущего символа sym, и переменную S, представляющую стек, в который 
будут помещаться номера позиций открывающих скобок. 

 Алгоритм обработки последовательности таков. Сначала счётчик 
позиций pos устанавливаем в ноль, создаём пустой стек (с помощью 
init_stack(S)) и открываем файл f для чтения. Пока не конец файла, 
повторяем следующие действия:   

1) считываем в sym очередной символ и увеличиваем pos на единицу; 
2) еcли текущий символ – открывающая скобка, то номер его позиции 

помещаем в стек (c помощью push(S, pos)); 
3) если текущий символ – закрывающая скобка, то это значит, что в 

переменной pos – номер позиции этой скобки, а на вершине стека – 
номер позиции соответствующей открывающей скобки; поэтому 
берём элемент из стека, помещая его во вспомогательную 
переменную i (с помощью pop(S,i)),  и печатаем пару (i, pos).  

В нашем алгоритме нет проверки стека на переполнение. В 
сбалансированном по скобкам тексте количество открывающих скобок не 
может превосходить половину длины текста. Поэтому размер стека должен 
быть не меньше, чем половина длины обрабатываемого текста, или входной 
текст должен иметь длину не более чем в два раза превосходящую размер 
стека.1  

                                                           
1 Если стек представлен не массивом, а  динамической цепочкой (см. задачу 10), то его размер 
заранее не ограничен. Стек может расти, пока у Паскаль-машины хватает ресурсов для создания 
нового динамического объекта – очередного звена стека. 

В алгоритме также нет проверки стека на пустоту перед тем, как взять 
элемент из стека. Эта проверка не нужна, так как в сбалансированной 
последовательности каждая открывающая скобка предшествует парной ей 
закрывающей скобке. Следовательно, номер любой открывающей скобки 
окажется в стеке раньше, чем произойдёт операция его извлечения на шаге (3). 
По достижении конца файла стек станет пустым, поскольку открывающих и 
закрывающих скобок в правильном входном тексте поровну. 

 Изобразим на рисунке последовательность состояний переменных f, 
output, pos, sym, S при обработке текста A+(45-F(X)*(B-C)). Файловые 
переменные будем изображать в виде ленты, разделённой на ячейки. В одной 
ячейке располагается одна компонента файла. Если файл открыт для чтения, то 
стрелка, указывающая снизу на ячейку, отмечает компоненту, которая будет 
считана при очередном обращении к процедуре read. Если стрелка указывает 
на место за последней ячейкой, то чтение очередной компоненты невозможно, 
так как достигнут конец файла (функция eof в этой ситуации возвращает 
значение «истина»). Стек представим в виде горизонтальной трубки, открытой с 
правого конца (вершина стека справа). Левый конец («дно» стека) отметим 
ломаной линией, символизирующей «пружину» для выталкивания элементов из 
стека.  

Перед чтением первого символа из f значение переменной sym не 
определено, pos=0, стек S и файл output пусты. 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

 ↑                  

 
output    

  
S        

 

 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

  ↑                 

 

 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

   ↑                

 

 

 

pos 0 sym

sym ’A’ pos 1 

sym ’+’ pos 2 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

    ↑               

 

 

 

 

 

 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

         ↑          

 

 

 
 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

          ↑         

 
 
 

 
 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

           ↑        

 
 
 

 
output ’8’ ’ ’ ’1’ ’0’ ’;’ ’ ’   

 
 
 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

             ↑      

S   3     

S   3 8    

S   3 8    

S   3     

sym ’(’ pos 3 

…

sym ’(’ pos 8 

sym ’X’ pos 9 

sym ’)’ pos 10 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

                 ↑  

 
 
 

 
output ’8’ ’ ’ ’1’ ’0’ ’;’ ’ ’ ’1’ ’2’ ’ ’ ’1’ ’6’ ’;’ ’ ’  

 
 
 

 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17  

f ’A’ ’+’ ’(’ ’4’ ’5’ ’-’ ’F’ ’(’ ’X’ ’)’ ’*’ ’(’ ’B’ ’-’ ’C’ ’)’ ’)’  

                  ↑ 
 

 
 

 output 

’8’ ’ ’ ’1’ ’0’ ’;’ ’ ’ ’1’ ’2’ ’ ’ ’1’ ’6’ ’;’ ’ ’ ’3’ ’ ’ ’1’ ’7’ ’;’ ’ ’ 
 
После того, как входная последовательность обработана (т.е. достигнут конец 
файла f) стек S пуст, в файле output результат – номера позиций парных 
скобок.  

Заметим, что текст может быть разбит на строки, т.е. кроме обычных 
символов (типа char) может встречаться специальный символ – маркер конца 
строки. При чтении этого маркера значение переменной pos изменять  не 
нужно, так он как не является «видимым» символом и не занимает отдельную 
позицию в исходной последовательности. Пропустить маркер конца строки и 
перейти к следующей строке можно с помощью оператора readln(f). 

procedure pairs(f:text); 
{ печатает пары номеров позиций открывающих и закрывающих   
  скобок из текста  f  в порядке возрастания номеров  для 
  закрывающих скобок} 
var sym: char; 
    pos,i: integer; 
    S: stack; 

S   3 12    

S   3     

S        

pos 12 sym ’(’

pos 16 sym ’)’

pos 17 sym ’)’

…

begin reset(f); 
      init_stack(S); pos:=0; 
  if not eof(f) then  
  repeat  
    if eoln(f) then readln(f) 
    else  
      begin read(f,sym); pos:=pos+1; 
        if sym = '(' then push(S, pos) 
          {номер позиции открывающей скобки в стек} 
        else if sym = ')' then 
          {номер позиции открывающей скобки,  
           соответствующей данной закрывающей, взять из  
           стека и напечатать пару номеров} 
          begin  pop(S,i);  
            write(i,' ',pos,'; ') 
          end; 
      end; 
  until  eof(f);  
  writeln; 
end {pairs}; 
 

Задача 10. Программа  на языке Турбо Паскаль. Программа запрашивает у 
пользователя имя текстового файла, в котором находится последовательность 
символов, сбалансированная по круглым скобкам, и выводит на экран номера 
парных скобок как в задаче 9. 

Решение 
Опишем в программе переменную name типа string и присвоим ей 

введённое пользователем имя внешнего файла, в котором хранится исходный 
текст. Опишем также переменную g типа text; c помощью процедуры assign 
свяжем файловую переменную g с внешним файлом name. Чтобы напечатать 
последовательность номеров парных скобок, воспользуемся процедурой pairs 
из задачи 9. Поскольку в pairs используется стек, мы должны реализовать в 
нашей программе соответствующие структуру данных и операции, т.е. описать 
тип stack и процедуры, необходимые для работы со стеком. Реализуем стек с 
помощью списка без заглавного звена. Процедура init_stack(S)делает 
список  S пустым; push(S,e)вставляет  в начало списка S элемент e – это 
операция «положить элемент в стек»; pop(S,e) удаляет из списка S первый 
элемент и присваивает его переменной e – это операция «вытолкнуть (взять) 
элемент из стека». После того, как работа с файлом закончена, его следует 
закрыть обращением к процедуре close. 

program parantheses(input, output); 
type elemtype=integer; 
     link=↑node; 
     node=record 
            elem:elemtype; 
            next:link 

          end; 
     stack=link; 
 
procedure  init_stack(var S:stack); 
begin 
  S:=nil 
end; 
 
procedure push(var S:stack; e:elemtype); 
var p:link; 
begin 
  new(p); 
  p↑.elem:=e; 
  p↑.next:=S; 
  S:=p 
end; 
 
procedure pop(var S:stack; var e:elemtype); 
var p:link; 
begin 
  p:=S; 
  S:=S↑.next; 
  e:=p↑.elem; 
  dispose(p) 
end; 
 
procedure pairs(var f:text); 
var sym  :char; 
    pos,i:integer; 
    S:stack; 
begin 
  reset(f); pos:=0; init_stack(S); 
  if not eof(f) then 
  repeat 
    if eoln(f) then readln(f) 
    else 
      begin read(f,sym); pos:=pos+1; 
        if      sym='(' then push(S,pos) 
        else if sym=')' then 
          begin pop(S,i); 
                write(i,'  ',pos,'; '); 
          end 
      end 
  until eof(f); 
  writeln 
end; 

var name:string; 
    g:text; 
begin 
  write('Введите имя файла: '); 
  readln(name); 
  assign(g,name); 
  pairs(g); 
  close(g); 
end. 
 
Сделаем два замечания к данной программе. Во-первых, в ней использованы 
средства языка Турбо Паскаль, которых нет в стандартном Паскале: assign, 
close, string. Во-вторых, программа будет корректно работать только с  
файлами, сбалансированными по скобкам. Если баланса скобок нет, во время 
выполнения может произойти ошибка, связанная с попыткой взять элемент из 
пустого стека. Примером файла, подходящего для обработки данной 
программой, является файл, содержащий её исходный текст. 

Задание практикума на ЭВМ 

В этом разделе приводятся варианты задания практикума на ЭВМ по 
теме «Динамические структуры данных». 

Постановка задачи 

Дан текст, состоящий из непустой последовательности слов из латинских 
букв, разделённых запятыми, за последним словом – точка; каждое слово 
состоит не более, чем из 10 символов. Требуется найти k – количество слов, 
которые удовлетворяют условию, заданному вариантом задания. В некоторых 
вариантах кроме текста задаётся ещё одна буква. 

Варианты 

Подсчитать количество слов, которые: 
1) имеют последней буквой заданную; 
2) содержат заданную букву ровно два раза; 
3) содержат заданную букву не менее двух раз; 
4) первой и последней буквой имеют одну и ту же букву; 
5) имеют длину не менее пяти букв; 
6) имеют первой буквой заданную и ещё хотя бы одно её вхождение; 
7) имеют последней буквой заданную и ещё хотя бы одно её вхождение; 
8) содержат заданную букву, но ни первой, ни последней; 
9) не имеют последней буквой заданную; 
10) имеют длину не более трёх букв; 

11) имеют первой и последней буквой одну и ту же заданную; 
12) не имеют первой и последней буквой одну и ту же заданную. 

Методические указания 

1. Вводить последовательность слов следует посимвольно. Для представления 
слов в программе использовать строковый тип подходящей длины. Слово 
располагать в начальных компонентах строки, не используемые компоненты 
заполнять пробелами. 
2. Для внутреннего представления текста использовать список с заглавным 
звеном. Звено списка состоит из трёх полей: первое – строка, в которой 
хранится слово, второе – длина данного слова, третье – указатель на следующее 
звено.  
3. Для вставки очередного слова в список определить отдельную процедуру; для 
подсчёта слов, удовлетворяющих условию, описать функцию. После построения 
списка напечатать слова, находящиеся в звеньях списка, используя процедуру 
print. 

Пример решения 

Пусть требуется найти, сколько в тексте слов длиной пять букв и в 
которых заданная буква встречается не менее двух раз. 

program sample(input, output); 
const n=10;  
      template=' '; 
type  line = packed array[1..n] of char; 
      size=1..10; 
      link= ↑node; 
      node = record 
               word: line; 
               length: size;  
               next: link  
             end; 
      list=link; 
var L: list; 
    p: link;  
    c: char; 
 word: line; 
  k,i: integer; 
 
procedure insert(p: link; word: line; length: size); 
{создаёт звено для слова word длины length и вставляет  
его в конец списка (после звена, на которое указывает p)} 
var q: link; 

begin new(q); 
  q↑.word:= word; 
  q↑.length:=length; 
  q↑.next:= nil{= p↑.next}; 
  p↑.next:= q 
end {insert}; 
 
procedure print(p: list); 
{печатает слова из списка с заглавным звеном} 
begin p:=p↑.next; 
  while p < > nil do   
    begin write(p↑.word,' '); 
          p:=p↑.next  
    end; 
  writeln; 
end {print}; 
 
function amount(p: list; letter: char; 
                         length: size):integer; 
{подсчитывает количество слов в списке, удовлетворяющих 
 условию: длина равна length и буква letter входит в  
 слово не менее двух раз} 
var k,m,i:integer; 
begin 
  k:=0;   p:= p↑.next; 
  while p < > nil do 
    begin 
      if p↑.length=length then 
        begin m:=0;  
          for i:=1 to length do  
            if p↑.word[i]= letter then m:= m+1; 
          if m>= 2 then k:= k+1  
        end; 
      p:=p↑.next 
    end; 
  amount:= k 
end {amount}; 
 
begin  {построение заглавного звена списка} 
  new(L); L↑.next:= nil;  
  writeln('Введите текст:'); 
  p:=L; {p установили на начало пустого списка} 
  repeat i:=0; word:= template; read(c); 
    {ввод очередного слова word из файла input} 
    repeat i:=i+1; word[i]:= c; read(c); 
    until (c = ',') or (c='.');  

    insert(p,word,i); 
    p:=p↑.next; {p указывает на последнее звено} 
  until c ='.'; 
  readln; 
  print(L); {печать введённого текста} 
  write('Задайте букву: '); 
  readln(c); 
{обработка списка и вывод результата} 
  writeln; 
  writeln('Результат: k=', amount(L,c,5)); 
  writeln('========================================'); 
end. 
 

Замечание. Идентификаторы  word и length являются стандартными 
именами в языке Турбо Паскаль. В нашей программе они переопределены и 
имеют другой смысл. 

